Require Export GeoCoq.Elements.OriginalProofs.lemma_twolines2.
Require Export GeoCoq.Elements.OriginalProofs.lemma_9_5b.
Require Export GeoCoq.Elements.OriginalProofs.lemma_9_5a.
Require Export GeoCoq.Elements.OriginalProofs.lemma_collinear5.
Require Export GeoCoq.Elements.OriginalProofs.proposition_10.

Section Euclid.

Context `{Ax:euclidean_neutral_ruler_compass}.

Lemma lemma_planeseparation :
   forall A B C D E,
   OS C D A B -> TS D A B E ->
   TS C A B E.
Proof.
intros.
rename_H H;
let Tf:=fresh in
assert (Tf:exists G H Q, (Col A B G /\ Col A B H /\ BetS C G Q /\ BetS D H Q /\ nCol A B C /\ nCol A B D)) by (conclude_def OS );destruct Tf as [G[H[Q]]];spliter.
assert (~ eq A B).
 {
 intro.
 assert (Col A B C) by (conclude_def Col ).
 contradict.
 }
assert (neq B A) by (conclude lemma_inequalitysymmetric).
let Tf:=fresh in
assert (Tf:exists W, (BetS D W E /\ Col A B W /\ nCol A B D)) by (conclude_def TS );destruct Tf as [W];spliter.
assert (neq C G) by (forward_using lemma_betweennotequal).
assert (neq G C) by (conclude lemma_inequalitysymmetric).
assert (neq G Q) by (forward_using lemma_betweennotequal).
assert (TS C A B E).
by cases on (Col C Q D \/ nCol C Q D).
{
 assert (~ ~ TS C A B E).
  {
  intro.
  assert (Col Q C D) by (forward_using lemma_collinearorder).
  assert (~ neq G H).
   {
   intro.
   assert (neq D Q) by (forward_using lemma_betweennotequal).
   assert (Col C G Q) by (conclude_def Col ).
   assert (Col D H Q) by (conclude_def Col ).
   assert (Col Q C G) by (forward_using lemma_collinearorder).
   assert (neq C Q) by (forward_using lemma_betweennotequal).
   assert (neq Q C) by (conclude lemma_inequalitysymmetric).
   assert (Col C G D) by (conclude lemma_collinear4).
   assert (Col D Q H) by (forward_using lemma_collinearorder).
   assert (Col D Q C) by (forward_using lemma_collinearorder).
   assert (Col Q H C) by (conclude lemma_collinear4).
   assert (Col Q C H) by (forward_using lemma_collinearorder).
   assert (~ Col C A B).
    {
    intro.
    assert (Col A B C) by (forward_using lemma_collinearorder).
    contradict.
    }
   assert (~ (Col C A B /\ Col D A B)).
    {
    intro.
    contradict.
    }
   assert (Col G A B) by (forward_using lemma_collinearorder).
   assert (Col H A B) by (forward_using lemma_collinearorder).
   assert (Col G C D) by (forward_using lemma_collinearorder).
   assert (Col Q D H) by (forward_using lemma_collinearorder).
   assert (Col Q D C) by (forward_using lemma_collinearorder).
   assert (neq D Q) by (forward_using lemma_betweennotequal).
   assert (neq Q D) by (conclude lemma_inequalitysymmetric).
   assert (Col D C H) by (conclude lemma_collinear4).
   assert (Col H C D) by (forward_using lemma_collinearorder).
   assert (~ eq C D).
    {
    intro.
    assert (TS C A B E) by (conclude cn_equalitysub).
    contradict.
    }
   assert (eq G H) by (conclude lemma_twolines2).
   contradict.
   }
  assert (BetS Q G C) by (conclude axiom_betweennesssymmetry).
  assert (BetS Q H D) by (conclude axiom_betweennesssymmetry).
  assert (BetS Q G D) by (conclude cn_equalitysub).
  assert (~ nCol E D G).
   {
   intro.
   assert (~ nCol E C G).
    {
    intro.
    assert (~ BetS G C D).
     {
     intro.
     assert (TS C A B E) by (conclude lemma_9_5b).
     contradict.
     }
    assert (~ BetS G D C).
     {
     intro.
     assert (~ Col G C E).
      {
      intro.
      assert (Col E C G) by (forward_using lemma_collinearorder).
      contradict.
      }
     assert (TS C A B E) by (conclude lemma_9_5a).
     contradict.
     }
    assert (eq C D) by (conclude lemma_outerconnectivity).
    assert (TS C A B E) by (conclude cn_equalitysub).
    contradict.
    }
   assert (Col C G E) by (forward_using lemma_collinearorder).
   assert (Col Q G D) by (conclude_def Col ).
   assert (Col Q G C) by (conclude_def Col ).
   assert (neq Q G) by (forward_using lemma_betweennotequal).
   assert (Col G D C) by (conclude lemma_collinear4).
   assert (Col C G D) by (forward_using lemma_collinearorder).
   assert (Col C G E) by (forward_using lemma_collinearorder).
   assert (neq G C) by (forward_using lemma_betweennotequal).
   assert (neq C G) by (conclude lemma_inequalitysymmetric).
   assert (Col G D E) by (conclude lemma_collinear4).
   assert (Col E D G) by (forward_using lemma_collinearorder).
   contradict.
   }
  assert (Col D G E) by (forward_using lemma_collinearorder).
  assert (Col D H Q) by (conclude_def Col ).
  assert (Col D G Q) by (conclude cn_equalitysub).
  assert (neq G D) by (forward_using lemma_betweennotequal).
  assert (neq D G) by (conclude lemma_inequalitysymmetric).
  assert (Col G E Q) by (conclude lemma_collinear4).
  assert (Col G E D) by (forward_using lemma_collinearorder).
  assert (Col D W E) by (conclude_def Col ).
  assert (Col D E W) by (forward_using lemma_collinearorder).
  assert (Col D E G) by (forward_using lemma_collinearorder).
  assert (neq D E) by (forward_using lemma_betweennotequal).
  assert (Col E W G) by (conclude lemma_collinear4).
  assert (Col E W D) by (forward_using lemma_collinearorder).
  assert (neq W E) by (forward_using lemma_betweennotequal).
  assert (neq E W) by (conclude lemma_inequalitysymmetric).
  assert (Col W G D) by (conclude lemma_collinear4).
  assert (Col B W G) by (conclude lemma_collinear4).
  assert (Col W G B) by (forward_using lemma_collinearorder).
  assert (Col B A W) by (forward_using lemma_collinearorder).
  assert (Col B A G) by (forward_using lemma_collinearorder).
  assert (Col A W G) by (conclude lemma_collinear4).
  assert (Col W G A) by (forward_using lemma_collinearorder).
  assert (~ neq W G).
   {
   intro.
   assert (Col G D B) by (conclude lemma_collinear4).
   assert (Col G B D) by (forward_using lemma_collinearorder).
   assert (Col G B A) by (forward_using lemma_collinearorder).
   assert (~ neq G B).
    {
    intro.
    assert (Col B D A) by (conclude lemma_collinear4).
    assert (Col A B D) by (forward_using lemma_collinearorder).
    contradict.
    }
   assert (Col G D A) by (conclude lemma_collinear4).
   assert (Col G A D) by (forward_using lemma_collinearorder).
   assert (Col G A B) by (forward_using lemma_collinearorder).
   assert (~ neq G A).
    {
    intro.
    assert (Col A D B) by (conclude lemma_collinear4).
    assert (Col A B D) by (forward_using lemma_collinearorder).
    contradict.
    }
   assert (eq A G) by (conclude lemma_equalitysymmetric).
   assert (eq B G) by (conclude lemma_equalitysymmetric).
   assert (eq A B) by (conclude cn_equalitytransitive).
   contradict.
   }
  assert (BetS D G E) by (conclude cn_equalitysub).
  assert (BetS E G D) by (conclude axiom_betweennesssymmetry).
  assert (~ BetS G D C).
   {
   intro.
   assert (BetS E G C) by (conclude lemma_3_7b).
   assert (BetS C G E) by (conclude axiom_betweennesssymmetry).
   assert (TS C A B E) by (conclude_def TS ).
   contradict.
   }
  assert (~ BetS G C D).
   {
   intro.
   assert (BetS E G C) by (conclude axiom_innertransitivity).
   assert (BetS C G E) by (conclude axiom_betweennesssymmetry).
   assert (TS C A B E) by (conclude_def TS ).
   contradict.
   }
  assert (~ BetS C G D).
   {
   intro.
   assert (BetS D G Q) by (conclude cn_equalitysub).
   assert (BetS Q G C) by (conclude axiom_betweennesssymmetry).
   assert (BetS D G C) by (conclude axiom_betweennesssymmetry).
   assert (neq G D) by (forward_using lemma_betweennotequal).
   assert (~ BetS G D C).
    {
    intro.
    assert (BetS C D G) by (conclude axiom_betweennesssymmetry).
    assert (BetS C G C) by (conclude lemma_3_7a).
    assert (neq C C) by (forward_using lemma_betweennotequal).
    assert (eq C C) by (conclude cn_equalityreflexive).
    contradict.
    }
   assert (~ BetS G C D).
    {
    intro.
    assert (BetS D G C) by (conclude axiom_betweennesssymmetry).
    assert (BetS C G C) by (conclude axiom_innertransitivity).
    assert (neq C C) by (forward_using lemma_betweennotequal).
    assert (eq C C) by (conclude cn_equalityreflexive).
    contradict.
    }
   assert (BetS Q G D) by (conclude axiom_betweennesssymmetry).
   assert (BetS Q G C) by (conclude axiom_betweennesssymmetry).
   assert (eq C D) by (conclude lemma_outerconnectivity).
   assert (TS C A B E) by (conclude cn_equalitysub).
   contradict.
   }
  assert (Col Q C D) by (forward_using lemma_collinearorder).
  assert (Col Q G C) by (conclude_def Col ).
  assert (Col Q C G) by (forward_using lemma_collinearorder).
  assert (neq Q C) by (forward_using lemma_betweennotequal).
  assert (Col C D G) by (conclude lemma_collinear4).
  assert (Col G C D) by (forward_using lemma_collinearorder).
  assert ((eq G C \/ eq G D \/ eq C D \/ BetS C G D \/ BetS G C D \/ BetS G D C)) by (conclude_def Col ).
  assert (TS C A B E).
  by cases on (eq G C \/ eq G D \/ eq C D \/ BetS C G D \/ BetS G C D \/ BetS G D C).
  {
   assert (~ ~ TS C A B E).
    {
    intro.
    assert (Col A B C) by (conclude cn_equalitysub).
    contradict.
    }
   close.
   }
  {
   assert (~ ~ TS C A B E).
    {
    intro.
    assert (Col A B D) by (conclude cn_equalitysub).
    contradict.
    }
   close.
   }
  {
   assert (~ ~ TS C A B E).
    {
    intro.
    assert (TS C A B E) by (conclude cn_equalitysub).
    contradict.
    }
   close.
   }
  {
   assert (~ ~ TS C A B E).
    {
    intro.
    contradict.
    }
   close.
   }
  {
   assert (~ ~ TS C A B E).
    {
    intro.
    contradict.
    }
   close.
   }
  {
   assert (~ ~ TS C A B E).
    {
    intro.
    contradict.
    }
   close.
   }
(** cases *)
  contradict.
  }
 close.
 }
{
 assert (~ ~ TS C A B E).
  {
  intro.
  assert (~ Col Q C D).
   {
   intro.
   assert (Col C Q D) by (forward_using lemma_collinearorder).
   contradict.
   }
  let Tf:=fresh in
  assert (Tf:exists F, (BetS C F H /\ BetS D F G)) by (conclude postulate_Pasch_inner);destruct Tf as [F];spliter.
  assert (BetS H F C) by (conclude axiom_betweennesssymmetry).
  assert (BetS G F D) by (conclude axiom_betweennesssymmetry).
  assert (~ Col E D G).
   {
   intro.
   assert (~ neq W G).
    {
    intro.
    assert (Col D E G) by (forward_using lemma_collinearorder).
    assert (Col D W E) by (conclude_def Col ).
    assert (Col B G W) by (conclude lemma_collinear4).
    assert (Col W B G) by (forward_using lemma_collinearorder).
    assert (Col W B A) by (forward_using lemma_collinearorder).
    assert (Col B G A).
    by cases on (eq W B \/ neq W B).
    {
     assert (Col B A G) by (forward_using lemma_collinearorder).
     assert (Col B A W) by (forward_using lemma_collinearorder).
     assert (Col A G W) by (conclude lemma_collinear4).
     assert (Col A G B) by (conclude cn_equalitysub).
     assert (Col B G A) by (forward_using lemma_collinearorder).
     close.
     }
    {
     assert (Col B G A) by (conclude lemma_collinear4).
     close.
     }
(** cases *)
    assert (Col G A B) by (forward_using lemma_collinearorder).
    assert (Col E D G) by (forward_using lemma_collinearorder).
    assert (Col E D W) by (forward_using lemma_collinearorder).
    assert (neq D E) by (forward_using lemma_betweennotequal).
    assert (neq E D) by (conclude lemma_inequalitysymmetric).
    assert (Col D G W) by (conclude lemma_collinear4).
    assert (Col G W D) by (forward_using lemma_collinearorder).
    assert (eq B B) by (conclude cn_equalityreflexive).
    assert (Col A B B) by (conclude_def Col ).
    assert (neq G W) by (conclude lemma_inequalitysymmetric).
    assert (Col G W B) by (conclude lemma_collinear5).
    assert (Col W D B) by (conclude lemma_collinear4).
    assert (Col W B D) by (forward_using lemma_collinearorder).
    assert (Col B A W) by (forward_using lemma_collinearorder).
    assert (Col B A G) by (forward_using lemma_collinearorder).
    assert (eq A A) by (conclude cn_equalityreflexive).
    assert (Col B A A) by (conclude_def Col ).
    assert (Col G W A) by (conclude lemma_collinear5).
    assert (Col W D A) by (conclude lemma_collinear4).
    assert (neq D W) by (forward_using lemma_betweennotequal).
    assert (neq W D) by (conclude lemma_inequalitysymmetric).
    assert (Col D B A) by (conclude lemma_collinear4).
    assert (Col A B D) by (forward_using lemma_collinearorder).
    contradict.
    }
   assert (BetS D G E) by (conclude cn_equalitysub).
   assert (BetS E G D) by (conclude axiom_betweennesssymmetry).
   assert (BetS G F D) by (conclude axiom_betweennesssymmetry).
   assert (BetS E G F) by (conclude axiom_innertransitivity).
   assert (~ Col H C E).
    {
    intro.
    assert (Col H E C) by (forward_using lemma_collinearorder).
    assert (Col E H C) by (forward_using lemma_collinearorder).
    assert (Col C F H) by (conclude_def Col ).
    assert (Col H C F) by (forward_using lemma_collinearorder).
    assert (Col H C E) by (forward_using lemma_collinearorder).
    assert (neq C H) by (forward_using lemma_betweennotequal).
    assert (neq H C) by (conclude lemma_inequalitysymmetric).
    assert (Col C F E) by (conclude lemma_collinear4).
    assert (Col E F C) by (forward_using lemma_collinearorder).
    assert (BetS D F E) by (conclude lemma_3_6b).
    assert (Col D F E) by (conclude_def Col ).
    assert (Col E F D) by (forward_using lemma_collinearorder).
    assert (neq F E) by (forward_using lemma_betweennotequal).
    assert (neq E F) by (conclude lemma_inequalitysymmetric).
    assert (Col F C D) by (conclude lemma_collinear4).
    assert (Col F D C) by (forward_using lemma_collinearorder).
    assert (Col D F G) by (conclude_def Col ).
    assert (Col F D G) by (forward_using lemma_collinearorder).
    assert (neq D F) by (forward_using lemma_betweennotequal).
    assert (neq F D) by (conclude lemma_inequalitysymmetric).
    assert (Col D C G) by (conclude lemma_collinear4).
    assert (Col G C D) by (forward_using lemma_collinearorder).
    assert (Col C G Q) by (conclude_def Col ).
    assert (Col G C Q) by (forward_using lemma_collinearorder).
    assert (neq C G) by (forward_using lemma_betweennotequal).
    assert (neq G C) by (conclude lemma_inequalitysymmetric).
    assert (Col C D Q) by (conclude lemma_collinear4).
    assert (Col Q C D) by (forward_using lemma_collinearorder).
    contradict.
    }
   let Tf:=fresh in
   assert (Tf:exists M, (BetS E M C /\ BetS H G M)) by (conclude postulate_Pasch_outer);destruct Tf as [M];spliter.
   assert (Col H G M) by (conclude_def Col ).
   assert (Col B H G) by (conclude lemma_collinear4).
   assert (Col H G B) by (forward_using lemma_collinearorder).
   assert (neq H G) by (forward_using lemma_betweennotequal).
   assert (Col G M B) by (conclude lemma_collinear4).
   assert (Col G B M) by (forward_using lemma_collinearorder).
   assert (Col G B A) by (forward_using lemma_collinearorder).
   assert (Col A B M).
   by cases on (eq B G \/ neq B G).
   {
    assert (Col B A H) by (forward_using lemma_collinearorder).
    assert (Col B A G) by (forward_using lemma_collinearorder).
    assert (Col A H G) by (conclude lemma_collinear4).
    assert (Col H G A) by (forward_using lemma_collinearorder).
    assert (Col G A M) by (conclude lemma_collinear4).
    assert (Col B A M) by (conclude cn_equalitysub).
    assert (Col A B M) by (forward_using lemma_collinearorder).
    close.
    }
   {
    assert (neq G B) by (conclude lemma_inequalitysymmetric).
    assert (Col B M A) by (conclude lemma_collinear4).
    assert (Col A B M) by (forward_using lemma_collinearorder).
    close.
    }
(** cases *)
   assert (BetS C M E) by (conclude axiom_betweennesssymmetry).
   assert (TS C A B E) by (conclude_def TS ).
   contradict.
   }
  assert (TS F A B E) by (conclude lemma_9_5b).
  assert (~ eq G H).
   {
   intro.
   assert (Col D H Q) by (conclude_def Col ).
   assert (Col C G Q) by (conclude_def Col ).
   assert (Col Q H D) by (forward_using lemma_collinearorder).
   assert (Col Q G C) by (forward_using lemma_collinearorder).
   assert (Col Q H C) by (conclude cn_equalitysub).
   assert (neq H Q) by (forward_using lemma_betweennotequal).
   assert (neq Q H) by (conclude lemma_inequalitysymmetric).
   assert (Col H D C) by (conclude lemma_collinear4).
   assert (Col H D Q) by (forward_using lemma_collinearorder).
   assert (neq D H) by (forward_using lemma_betweennotequal).
   assert (neq H D) by (conclude lemma_inequalitysymmetric).
   assert (Col D C Q) by (conclude lemma_collinear4).
   assert (Col C Q D) by (forward_using lemma_collinearorder).
   contradict.
   }
  assert (~ Col H C E).
   {
   intro.
   let Tf:=fresh in
   assert (Tf:exists K, (BetS G K H /\ Cong K G K H)) by (conclude proposition_10);destruct Tf as [K];spliter.
   assert (~ Col G D E).
    {
    intro.
    assert (Col E D G) by (forward_using lemma_collinearorder).
    contradict.
    }
   assert (BetS E W D) by (conclude axiom_betweennesssymmetry).
   assert (BetS G F D) by (conclude axiom_betweennesssymmetry).
   let Tf:=fresh in
   assert (Tf:exists J, (BetS E J F /\ BetS G J W)) by (conclude postulate_Pasch_inner);destruct Tf as [J];spliter.
   assert (Col G J W) by (conclude_def Col ).
   assert (Col E F J) by (conclude_def Col ).
   assert (Col E C H) by (forward_using lemma_collinearorder).
   assert (Col F E J) by (forward_using lemma_collinearorder).
   assert (Col C F H) by (conclude_def Col ).
   assert (Col C H F) by (forward_using lemma_collinearorder).
   assert (Col C H E) by (forward_using lemma_collinearorder).
   assert (neq C H) by (forward_using lemma_betweennotequal).
   assert (Col H F E) by (conclude lemma_collinear4).
   assert (Col F E H) by (forward_using lemma_collinearorder).
   assert (neq E F) by (forward_using lemma_betweennotequal).
   assert (neq F E) by (conclude lemma_inequalitysymmetric).
   assert (Col E J H) by (conclude lemma_collinear4).
   assert (Col E H J) by (forward_using lemma_collinearorder).
   assert (Col G W J) by (forward_using lemma_collinearorder).
   assert (eq H H) by (conclude cn_equalityreflexive).
   assert (Col E H H) by (conclude_def Col ).
   assert (Col G H W) by (conclude lemma_collinear5).
   assert (Col G W H) by (forward_using lemma_collinearorder).
   assert (~ eq G W).
    {
    intro.
    assert (Col D W E) by (conclude_def Col ).
    assert (Col D G E) by (conclude cn_equalitysub).
    assert (Col E D G) by (forward_using lemma_collinearorder).
    contradict.
    }
   assert (Col E J H) by (conclude lemma_collinear4).
   assert (Col W J H) by (conclude lemma_collinear4).
   assert (Col J H E) by (forward_using lemma_collinearorder).
   assert (Col J H W) by (forward_using lemma_collinearorder).
   assert (~ eq H W).
    {
    intro.
    assert (Col D W E) by (conclude_def Col ).
    assert (Col D H E) by (conclude cn_equalitysub).
    assert (Col D H Q) by (conclude_def Col ).
    assert (neq D H) by (forward_using lemma_betweennotequal).
    assert (Col H E Q) by (conclude lemma_collinear4).
    assert (Col H E C) by (forward_using lemma_collinearorder).
    assert (neq W E) by (forward_using lemma_betweennotequal).
    assert (neq H E) by (conclude cn_equalitysub).
    assert (Col E Q C) by (conclude lemma_collinear4).
    assert (Col E C Q) by (forward_using lemma_collinearorder).
    assert (Col E C H) by (forward_using lemma_collinearorder).
    assert (~ neq E C).
     {
     intro.
     assert (Col C Q H) by (conclude lemma_collinear4).
     assert (Col H Q C) by (forward_using lemma_collinearorder).
     assert (Col H Q D) by (forward_using lemma_collinearorder).
     assert (neq H Q) by (forward_using lemma_betweennotequal).
     assert (Col Q C D) by (conclude lemma_collinear4).
     contradict.
     }
    assert (Col E W D) by (forward_using lemma_collinearorder).
    assert (Col C W D) by (conclude cn_equalitysub).
    assert (Col C H D) by (conclude cn_equalitysub).
    assert (Col D H C) by (forward_using lemma_collinearorder).
    assert (Col D H Q) by (conclude_def Col ).
    assert (neq D H) by (forward_using lemma_betweennotequal).
    assert (Col H C Q) by (conclude lemma_collinear4).
    assert (Col H Q C) by (forward_using lemma_collinearorder).
    assert (Col H Q D) by (forward_using lemma_collinearorder).
    assert (neq H Q) by (forward_using lemma_betweennotequal).
    assert (Col Q C D) by (conclude lemma_collinear4).
    assert (Col C Q D) by (forward_using lemma_collinearorder).
    contradict.
    }
   assert (~ neq J H).
    {
    intro.
    assert (Col H E W) by (conclude lemma_collinear4).
    assert (Col H W E) by (forward_using lemma_collinearorder).
    assert (Col H W G) by (forward_using lemma_collinearorder).
    assert (Col W E G) by (conclude lemma_collinear4).
    assert (Col D W E) by (conclude_def Col ).
    assert (Col W E D) by (forward_using lemma_collinearorder).
    assert (neq W E) by (forward_using lemma_betweennotequal).
    assert (Col E G D) by (conclude lemma_collinear4).
    assert (Col E D G) by (forward_using lemma_collinearorder).
    contradict.
    }
   assert (BetS E H F) by (conclude cn_equalitysub).
   assert (BetS F H E) by (conclude axiom_betweennesssymmetry).
   assert (BetS C H E) by (conclude lemma_3_7a).
   assert (TS C A B E) by (conclude_def TS ).
   contradict.
   }
  assert (TS C A B E) by (conclude lemma_9_5a).
  contradict.
  }
 close.
 }
(** cases *)
close.
Qed.

End Euclid.


(** * Hoare: Hoare Logic, Part I *)
Add LoadPath "~/Desktop/CS386L-Programming-Language/plf".
Set Warnings "-notation-overridden,-parsing".
Require Import Coq.Bool.Bool.
Require Import Coq.Arith.Arith.
Require Import Coq.Arith.EqNat.
Require Import Coq.omega.Omega.
Require Import Imp.
Require Import Maps.

(** In the final chaper of _Logical Foundations_ (_Software
    Foundations_, volume 1), we began applying the mathematical tools
    developed in the first part of the course to studying the theory
    of a small programming language, Imp.

    - We defined a type of _abstract syntax trees_ for Imp, together
      with an _evaluation relation_ (a partial function on states)
      that specifies the _operational semantics_ of programs.

      The language we defined, though small, captures some of the key
      features of full-blown languages like C, C++, and Java,
      including the fundamental notion of mutable state and some
      common control structures.

    - We proved a number of _metatheoretic properties_ -- "meta" in
      the sense that they are properties of the language as a whole,
      rather than of particular programs in the language.  These
      included:

        - determinism of evaluation

        - equivalence of some different ways of writing down the
          definitions (e.g., functional and relational definitions of
          arithmetic expression evaluation)

        - guaranteed termination of certain classes of programs

        - correctness (in the sense of preserving meaning) of a number
          of useful program transformations

        - behavioral equivalence of programs (in the [Equiv]
          chapter). *)

(** If we stopped here, we would already have something useful: a set
    of tools for defining and discussing programming languages and
    language features that are mathematically precise, flexible, and
    easy to work with, applied to a set of key properties.  All of
    these properties are things that language designers, compiler
    writers, and users might care about knowing.  Indeed, many of them
    are so fundamental to our understanding of the programming
    languages we deal with that we might not consciously recognize
    them as "theorems."  But properties that seem intuitively obvious
    can sometimes be quite subtle (sometimes also subtly wrong!).

    We'll return to the theme of metatheoretic properties of whole
    languages later in this volume when we discuss _types_ and _type
    soundness_.  In this chapter, though, we turn to a different set
    of issues.

    Our goal is to carry out some simple examples of _program
    verification_ -- i.e., to use the precise definition of Imp to
    prove formally that particular programs satisfy particular
    specifications of their behavior.  We'll develop a reasoning
    system called _Floyd-Hoare Logic_ -- often shortened to just
    _Hoare Logic_ -- in which each of the syntactic constructs of Imp
    is equipped with a generic "proof rule" that can be used to reason
    compositionally about the correctness of programs involving this
    construct.

    Hoare Logic originated in the 1960s, and it continues to be the
    subject of intensive research right up to the present day.  It
    lies at the core of a multitude of tools that are being used in
    academia and industry to specify and verify real software systems.

    Hoare Logic combines two beautiful ideas: a natural way of writing
    down _specifications_ of programs, and a _compositional proof
    technique_ for proving that programs are correct with respect to
    such specifications -- where by "compositional" we mean that the
    structure of proofs directly mirrors the structure of the programs
    that they are about. *)

(** This chapter:
      - A systematic method for reasoning about the _functional
        correctness_ of programs in Imp

    Goals:
      - a natural notation for _program specifications_ and
      - a _compositional_ proof technique for program correctness

    Plan:
      - specifications (assertions / Hoare triples)
      - proof rules
      - loop invariants
      - decorated programs
      - examples *)

(* ################################################################# *)
(** * Assertions *)

(** To talk about specifications of programs, the first thing we
    need is a way of making _assertions_ about properties that hold at
    particular points during a program's execution -- i.e., claims
    about the current state of the memory when execution reaches that
    point.  Formally, an assertion is just a family of propositions
    indexed by a [state]. *)

Definition Assertion := state -> Prop.

(** **** Exercise: 1 star, optional (assertions)  *)
(** Paraphrase the following assertions in English (or your favorite
    natural language). *)

Module ExAssertions.
Definition as1 : Assertion := fun st => st X = 3.
Definition as2 : Assertion := fun st => st X <= st Y.
Definition as3 : Assertion :=
  fun st => st X = 3 \/ st X <= st Y.
Definition as4 : Assertion :=
  fun st => st Z * st Z <= st X /\
            ~ (((S (st Z)) * (S (st Z))) <= st X).
Definition as5 : Assertion := fun st => True.
Definition as6 : Assertion := fun st => False.
(*
as1: the value of X is 3
as2: the value of X ls less or equal than the value of Y
as3: as1 is true or as2 is true
as4: z^2<=x and(z+1)*(z+1)>x
as5: always true
as6: always false
 *)
End ExAssertions.
(** [] *)

(** This way of writing assertions can be a little bit heavy,
    for two reasons: (1) every single assertion that we ever write is
    going to begin with [fun st => ]; and (2) this state [st] is the
    only one that we ever use to look up variables in assertions (we
    will never need to talk about two different memory states at the
    same time).  For discussing examples informally, we'll adopt some
    simplifying conventions: we'll drop the initial [fun st =>], and
    we'll write just [X] to mean [st X].  Thus, instead of writing *)
(**

      fun st => (st Z) * (st Z) <= m /\
                ~ ((S (st Z)) * (S (st Z)) <= m)

    we'll write just

      Z * Z <= m /\ ~((S Z) * (S Z) <= m).
*)

(** This example also illustrates a convention that we'll use
    throughout the Hoare Logic chapters: in informal assertions,
    capital letters like [X], [Y], and [Z] are Imp variables, while
    lowercase letters like [x], [y], [m], and [n] are ordinary Coq
    variables (of type [nat]).  This is why, when translating from
    informal to formal, we replace [X] with [st X] but leave [m]
    alone. *)

(** Given two assertions [P] and [Q], we say that [P] _implies_ [Q],
    written [P ->> Q], if, whenever [P] holds in some state [st], [Q]
    also holds. *)

Definition assert_implies (P Q : Assertion) : Prop :=
  forall st, P st -> Q st.

Notation "P ->> Q" := (assert_implies P Q)
                      (at level 80) : hoare_spec_scope.
Open Scope hoare_spec_scope.

(** (The [hoare_spec_scope] annotation here tells Coq that this
    notation is not global but is intended to be used in particular
    contexts.  The [Open Scope] tells Coq that this file is one such
    context.) *)

(** We'll also want the "iff" variant of implication between
    assertions: *)

Notation "P <<->> Q" :=
  (P ->> Q /\ Q ->> P) (at level 80) : hoare_spec_scope.

(* ################################################################# *)
(** * Hoare Triples *)

(** Next, we need a way of making formal claims about the
    behavior of commands. *)

(** In general, the behavior of a command is to transform one state to
    another, so it is natural to express claims about commands in
    terms of assertions that are true before and after the command
    executes:

      - "If command [c] is started in a state satisfying assertion
        [P], and if [c] eventually terminates in some final state,
        then this final state will satisfy the assertion [Q]."

    Such a claim is called a _Hoare Triple_.  The assertion [P] is
    called the _precondition_ of [c], while [Q] is the
    _postcondition_.  *)

(** Formally: *)

Definition hoare_triple
           (P:Assertion) (c:com) (Q:Assertion) : Prop :=
  forall st st',
     c / st \\ st'  ->
     P st  ->
     Q st'.

(** Since we'll be working a lot with Hoare triples, it's useful to
    have a compact notation:

       {{P}} c {{Q}}.
*)
(** (The traditional notation is [{P} c {Q}], but single braces
    are already used for other things in Coq.)  *)

Notation "{{ P }}  c  {{ Q }}" :=
  (hoare_triple P c Q) (at level 90, c at next level)
  : hoare_spec_scope.

(** **** Exercise: 1 star, optional (triples)  *)
(** Paraphrase the following Hoare triples in English.

   1) {{True}} c {{X = 5}}

   2) {{X = m}} c {{X = m + 5)}}

   3) {{X <= Y}} c {{Y <= X}}

   4) {{True}} c {{False}}

   5) {{X = m}}
      c
      {{Y = real_fact m}}

   6) {{X = m}}
      c
      {{(Z * Z) <= m /\ ~ (((S Z) * (S Z)) <= m)}}
*)

(** [] *)

(** **** Exercise: 1 star, optional (valid_triples)  *)
(** Which of the following Hoare triples are _valid_ -- i.e., the
    claimed relation between [P], [c], and [Q] is true?

   1) {{True}} X ::= 5 {{X = 5}}

   2) {{X = 2}} X ::= X + 1 {{X = 3}}

   3) {{True}} X ::= 5; Y ::= 0 {{X = 5}}

   4) {{X = 2 /\ X = 3}} X ::= 5 {{X = 0}}

   5) {{True}} SKIP {{False}}

   6) {{False}} SKIP {{True}}

   7) {{True}} WHILE true DO SKIP END {{False}}

   8) {{X = 0}}
        WHILE X = 0 DO X ::= X + 1 END
      {{X = 1}}

   9) {{X = 1}}
        WHILE !(X = 0) DO X ::= X + 1 END
      {{X = 100}}
*)
(** [] *)

(** To get us warmed up for what's coming, here are two simple facts
    about Hoare triples.  (Make sure you understand what they mean.) *)

Theorem hoare_post_true : forall (P Q : Assertion) c,
  (forall st, Q st) ->
  {{P}} c {{Q}}.
Proof.
  intros P Q c H. unfold hoare_triple.
  intros st st' Heval HP.
  apply H.  Qed.

Theorem hoare_pre_false : forall (P Q : Assertion) c,
  (forall st, ~(P st)) ->
  {{P}} c {{Q}}.
Proof.
  intros P Q c H. unfold hoare_triple.
  intros st st' Heval HP.
  unfold not in H. apply H in HP.
  inversion HP.  Qed.

(* ################################################################# *)
(** * Proof Rules *)

(** The goal of Hoare logic is to provide a _compositional_
    method for proving the validity of specific Hoare triples.  That
    is, we want the structure of a program's correctness proof to
    mirror the structure of the program itself.  To this end, in the
    sections below, we'll introduce a rule for reasoning about each of
    the different syntactic forms of commands in Imp -- one for
    assignment, one for sequencing, one for conditionals, etc. -- plus
    a couple of "structural" rules for gluing things together.  We
    will then be able to prove programs correct using these proof
    rules, without ever unfolding the definition of [hoare_triple]. *)

(* ================================================================= *)
(** ** Assignment *)

(** The rule for assignment is the most fundamental of the Hoare logic
    proof rules.  Here's how it works.

    Consider this valid Hoare triple:

       {{ Y = 1 }}  X ::= Y  {{ X = 1 }}

    In English: if we start out in a state where the value of [Y]
    is [1] and we assign [Y] to [X], then we'll finish in a
    state where [X] is [1].
    That is, the property of being equal to [1] gets transferred
    from [Y] to [X]. *)

(** Similarly, in

       {{ Y + Z = 1 }}  X ::= Y + Z  {{ X = 1 }}

    the same property (being equal to one) gets transferred to
    [X] from the expression [Y + Z] on the right-hand side of
    the assignment. *)

(** More generally, if [a] is _any_ arithmetic expression, then

       {{ a = 1 }}  X ::= a {{ X = 1 }}

    is a valid Hoare triple. *)

(** Even more generally, to conclude that an arbitrary assertion [Q]
    holds after [X ::= a], we need to assume that [Q] holds before [X
    ::= a], but _with all occurrences of_ [X] replaced by [a] in
    [Q]. This leads to the Hoare rule for assignment

      {{ Q [X |-> a] }} X ::= a {{ Q }}

    where "[Q [X |-> a]]" is pronounced "[Q] where [a] is substituted
    for [X]". *)

(** For example, these are valid applications of the assignment
    rule:

      {{ (X <= 5) [X |-> X + 1]
         i.e., X + 1 <= 5 }}
      X ::= X + 1
      {{ X <= 5 }}

      {{ (X = 3) [X |-> 3]
         i.e., 3 = 3}}
      X ::= 3
      {{ X = 3 }}

      {{ (0 <= X /\ X <= 5) [X |-> 3]
         i.e., (0 <= 3 /\ 3 <= 5)}}
      X ::= 3
      {{ 0 <= X /\ X <= 5 }}
*)

(** To formalize the rule, we must first formalize the idea of
    "substituting an expression for an Imp variable in an assertion",
    which we refer to as assertion substitution, or [assn_sub].  That
    is, given a proposition [P], a variable [X], and an arithmetic
    expression [a], we want to derive another proposition [P'] that is
    just the same as [P] except that [P'] should mention [a] wherever
    [P] mentions [X]. *)

(** Since [P] is an arbitrary Coq assertion, we can't directly "edit"
    its text.  However, we can achieve the same effect by evaluating
    [P] in an updated state: *)

Definition assn_sub X a P : Assertion :=
  fun (st : state) =>
    P (st & { X  --> aeval st a }).

Notation "P [ X |-> a ]" := (assn_sub X a P) (at level 10).

(** That is, [P [X |-> a]] stands for an assertion -- let's call it [P'] --
    that is just like [P] except that, wherever [P] looks up the
    variable [X] in the current state, [P'] instead uses the value
    of the expression [a]. *)

(** To see how this works, let's calculate what happens with a couple
    of examples.  First, suppose [P'] is [(X <= 5) [X |-> 3]] -- that
    is, more formally, [P'] is the Coq expression

    fun st =>
      (fun st' => st' X <= 5)
      (st & { X --> aeval st 3 }),

    which simplifies to

    fun st =>
      (fun st' => st' X <= 5)
      (st & { X --> 3 })

    and further simplifies to

    fun st =>
      ((st & { X --> 3 }) X) <= 5

    and finally to

    fun st =>
      3 <= 5.

    That is, [P'] is the assertion that [3] is less than or equal to
    [5] (as expected). *)

(** For a more interesting example, suppose [P'] is [(X <= 5) [X |->
    X+1]].  Formally, [P'] is the Coq expression

    fun st =>
      (fun st' => st' X <= 5)
      (st & { X --> aeval st (X+1) }),

    which simplifies to

    fun st =>
      (st & { X --> aeval st (X+1) }) X <= 5

    and further simplifies to

    fun st =>
      (aeval st (X+1)) <= 5.

    That is, [P'] is the assertion that [X+1] is at most [5].
*)

(** Now, using the concept of substitution, we can give the precise
    proof rule for assignment:

      ------------------------------ (hoare_asgn)
      {{Q [X |-> a]}} X ::= a {{Q}}
*)

(** We can prove formally that this rule is indeed valid. *)

Theorem hoare_asgn : forall Q X a,
  {{Q [X |-> a]}} (X ::= a) {{Q}}.
Proof.
  unfold hoare_triple.
  intros Q X a st st' HE HQ.
  inversion HE. subst.
  unfold assn_sub in HQ. assumption.  Qed.

(** Here's a first formal proof using this rule. *)

Example assn_sub_example :
  {{(fun st => st X < 5) [X |-> X+1]}}
  (X ::= X+1)
  {{fun st => st X < 5}}.
Proof.
  (* WORKED IN CLASS *)
  apply hoare_asgn.  Qed.

(** Of course, what would be even more helpful is to prove this
    simpler triple:

      {{X < 4}} (X ::= X+1) {{X < 5}}

   We will see how to do so in the next section. *)

(** **** Exercise: 2 stars (hoare_asgn_examples)  *)
(** Translate these informal Hoare triples...

    1) {{ (X <= 10) [X |-> 2 * X] }}
       X ::= 2 * X
       {{ X <= 10 }}

    2) {{ (0 <= X /\ X <= 5) [X |-> 3] }}
       X ::= 3
       {{ 0 <= X /\ X <= 5 }}

   ...into formal statements (use the names [assn_sub_ex1]
   and [assn_sub_ex2]) and use [hoare_asgn] to prove them. *)

Theorem assn_sub_ex1:
  {{ (fun st=>st X<=10) [X |-> 2*X]}}
  (X ::= 2 *X)
  {{ (fun st=>st X<=10) }}.
Proof. apply hoare_asgn. Qed.

Theorem assn_sub_ex2:
  {{ (fun st=>0<=st X /\ st X<=5) [X |-> 3] }}
  (X ::= 3)
  {{ (fun st=>0<=st X /\ st X<=5) }}.
Proof.
  apply hoare_asgn.
Qed.
(** [] *)

(** **** Exercise: 2 stars, recommended (hoare_asgn_wrong)  *)
(** The assignment rule looks backward to almost everyone the first
    time they see it.  If it still seems puzzling, it may help
    to think a little about alternative "forward" rules.  Here is a
    seemingly natural one:

      ------------------------------ (hoare_asgn_wrong)
      {{ True }} X ::= a {{ X = a }}

    Give a counterexample showing that this rule is incorrect and
    argue informally that it is really a counterexample.  (Hint:
    The rule universally quantifies over the arithmetic expression
    [a], and your counterexample needs to exhibit an [a] for which
    the rule doesn't work.) *)

(* let a=X+1, then X::=a means increase X by 1.
   But the later assertion {{X=a}} -> {{X==X+1}} will never be true for nats.
 *)
(** [] *)

Local Close Scope aexp_scope.

(** **** Exercise: 3 stars, advanced (hoare_asgn_fwd)  *)
(** However, by using a _parameter_ [m] (a Coq number) to remember the
    original value of [X] we can define a Hoare rule for assignment
    that does, intuitively, "work forwards" rather than backwards.

       ------------------------------------------ (hoare_asgn_fwd)
       {{fun st => P st /\ st X = m}}
         X ::= a
       {{fun st => P st' /\ st X = aeval st' a }}
       (where st' = st & { X --> m })

    Note that we use the original value of [X] to reconstruct the
    state [st'] before the assignment took place. Prove that this rule
    is correct.  (Also note that this rule is more complicated than
    [hoare_asgn].)
*)

Theorem hoare_asgn_fwd :
  forall m a P,
  {{fun st => P st /\ st X = m}}
    X ::= a
  {{fun st => P (st & { X --> m })
            /\ st X = aeval  (st & { X --> m }) a }}.
Proof.
  intros. unfold hoare_triple. intros.
  inversion H; subst. destruct H0.
  assert ((st & {X --> aeval st a; X --> m})=st & {X-->m}).
  apply t_update_shadow.
  assert ((st & {X-->m})=st).
  rewrite<-H1.  apply t_update_same.
  split.
  - rewrite H2.
    rewrite H3. apply H0.
  - assert (((st & {X --> aeval st a}) X)=aeval st a).
    apply t_update_eq. rewrite H4.
    rewrite H2. rewrite H3. reflexivity.
Qed.
(** [] *)

(** **** Exercise: 2 stars, advanced, optional (hoare_asgn_fwd_exists)  *)
(** Another way to define a forward rule for assignment is to
    existentially quantify over the previous value of the assigned
    variable.  Prove that it is correct.

      ------------------------------------ (hoare_asgn_fwd_exists)
      {{fun st => P st}}
        X ::= a
      {{fun st => exists m, P (st & { X --> m }) /\
                     st X = aeval (st & { X --> m }) a }}
*)

Theorem hoare_asgn_fwd_exists :
  forall a P,
  {{fun st => P st}}
    X ::= a
  {{fun st => exists m, P (st & { X --> m }) /\
                st X = aeval (st & { X --> m }) a }}.
Proof.
  intros a P.
  (* FILL IN HERE *) Admitted.
(** [] *)

(* ================================================================= *)
(** ** Consequence *)

(** Sometimes the preconditions and postconditions we get from the
    Hoare rules won't quite be the ones we want in the particular
    situation at hand -- they may be logically equivalent but have a
    different syntactic form that fails to unify with the goal we are
    trying to prove, or they actually may be logically weaker (for
    preconditions) or stronger (for postconditions) than what we need.

    For instance, while

      {{(X = 3) [X |-> 3]}} X ::= 3 {{X = 3}},

    follows directly from the assignment rule,

      {{True}} X ::= 3 {{X = 3}}

    does not.  This triple is valid, but it is not an instance of
    [hoare_asgn] because [True] and [(X = 3) [X |-> 3]] are not
    syntactically equal assertions.  However, they are logically
    _equivalent_, so if one triple is valid, then the other must
    certainly be as well.  We can capture this observation with the
    following rule:

                {{P'}} c {{Q}}
                  P <<->> P'
         -----------------------------   (hoare_consequence_pre_equiv)
                {{P}} c {{Q}}
*)

(** Taking this line of thought a bit further, we can see that
    strengthening the precondition or weakening the postcondition of a
    valid triple always produces another valid triple. This
    observation is captured by two _Rules of Consequence_.

                {{P'}} c {{Q}}
                   P ->> P'
         -----------------------------   (hoare_consequence_pre)
                {{P}} c {{Q}}

                {{P}} c {{Q'}}
                  Q' ->> Q
         -----------------------------    (hoare_consequence_post)
                {{P}} c {{Q}}
*)

(** Here are the formal versions: *)

Theorem hoare_consequence_pre : forall (P P' Q : Assertion) c,
  {{P'}} c {{Q}} ->
  P ->> P' ->
  {{P}} c {{Q}}.
Proof.
  intros P P' Q c Hhoare Himp.
  intros st st' Hc HP. apply (Hhoare st st').
  assumption. apply Himp. assumption. Qed.

Theorem hoare_consequence_post : forall (P Q Q' : Assertion) c,
  {{P}} c {{Q'}} ->
  Q' ->> Q ->
  {{P}} c {{Q}}.
Proof.
  intros P Q Q' c Hhoare Himp.
  intros st st' Hc HP.
  apply Himp.
  apply (Hhoare st st').
  assumption. assumption. Qed.

(** For example, we can use the first consequence rule like this:

                {{ True }} ->>
                {{ 1 = 1 }}
    X ::= 1
                {{ X = 1 }}

    Or, formally... *)

Example hoare_asgn_example1 :
  {{fun st => True}} (X ::= 1) {{fun st => st X = 1}}.
Proof.
  (* WORKED IN CLASS *)
  apply hoare_consequence_pre
    with (P' := (fun st => st X = 1) [X |-> 1]).
  apply hoare_asgn.
  intros st H. unfold assn_sub, t_update. simpl. reflexivity.
Qed.

(** We can also use it to prove the example mentioned earlier.

		{{ X < 4 }} ->>
		{{ (X < 5)[X |-> X+1] }}
    X ::= X + 1
	        {{ X < 5 }}

   Or, formally ... *)

Example assn_sub_example2 :
  {{(fun st => st X < 4)}}
  (X ::= X+1)
  {{fun st => st X < 5}}.
Proof.
  (* WORKED IN CLASS *)
  apply hoare_consequence_pre
    with (P' := (fun st => st X < 5) [X |-> X+1]).
  apply hoare_asgn.
  intros st H. unfold assn_sub, t_update. simpl. omega.
Qed.

(** Finally, for convenience in proofs, here is a combined rule of
    consequence that allows us to vary both the precondition and the
    postcondition in one go.

                {{P'}} c {{Q'}}
                   P ->> P'
                   Q' ->> Q
         -----------------------------   (hoare_consequence)
                {{P}} c {{Q}}
*)

Theorem hoare_consequence : forall (P P' Q Q' : Assertion) c,
  {{P'}} c {{Q'}} ->
  P ->> P' ->
  Q' ->> Q ->
  {{P}} c {{Q}}.
Proof.
  intros P P' Q Q' c Hht HPP' HQ'Q.
  apply hoare_consequence_pre with (P' := P').
  apply hoare_consequence_post with (Q' := Q').
  assumption. assumption. assumption.  Qed.

(* ================================================================= *)
(** ** Digression: The [eapply] Tactic *)

(** This is a good moment to take another look at the [eapply] tactic,
    which we introduced briefly in the [Auto] chapter of
    _Logical Foundations_.

    We had to write "[with (P' := ...)]" explicitly in the proof of
    [hoare_asgn_example1] and [hoare_consequence] above, to make sure
    that all of the metavariables in the premises to the
    [hoare_consequence_pre] rule would be set to specific
    values.  (Since [P'] doesn't appear in the conclusion of
    [hoare_consequence_pre], the process of unifying the conclusion
    with the current goal doesn't constrain [P'] to a specific
    assertion.)

    This is annoying, both because the assertion is a bit long and
    also because, in [hoare_asgn_example1], the very next thing we are
    going to do -- applying the [hoare_asgn] rule -- will tell us
    exactly what it should be!  We can use [eapply] instead of [apply]
    to tell Coq, essentially, "Be patient: The missing part is going
    to be filled in later in the proof." *)

Example hoare_asgn_example1' :
  {{fun st => True}}
  (X ::= 1)
  {{fun st => st X = 1}}.
Proof.
  eapply hoare_consequence_pre.
  apply hoare_asgn.
  intros st H.  reflexivity.  Qed.

(** In general, [eapply H] tactic works just like [apply H] except
    that, instead of failing if unifying the goal with the conclusion
    of [H] does not determine how to instantiate all of the variables
    appearing in the premises of [H], [eapply H] will replace these
    variables with _existential variables_ (written [?nnn]), which
    function as placeholders for expressions that will be
    determined (by further unification) later in the proof. *)

(** In order for [Qed] to succeed, all existential variables need to
    be determined by the end of the proof. Otherwise Coq
    will (rightly) refuse to accept the proof. Remember that the Coq
    tactics build proof objects, and proof objects containing
    existential variables are not complete. *)

Lemma silly1 : forall (P : nat -> nat -> Prop) (Q : nat -> Prop),
  (forall x y : nat, P x y) ->
  (forall x y : nat, P x y -> Q x) ->
  Q 42.
Proof.
  intros P Q HP HQ. eapply HQ. apply HP.

(** Coq gives a warning after [apply HP].  ("All the remaining goals
    are on the shelf," means that we've finished all our top-level
    proof obligations but along the way we've put some aside to be
    done later, and we have not finished those.)  Trying to close the
    proof with [Qed] gives an error. *)
Abort.

(** An additional constraint is that existential variables cannot be
    instantiated with terms containing ordinary variables that did not
    exist at the time the existential variable was created.  (The
    reason for this technical restriction is that allowing such
    instantiation would lead to inconsistency of Coq's logic.) *)

Lemma silly2 :
  forall (P : nat -> nat -> Prop) (Q : nat -> Prop),
  (exists y, P 42 y) ->
  (forall x y : nat, P x y -> Q x) ->
  Q 42.
Proof.
  intros P Q HP HQ. eapply HQ. destruct HP as [y HP'].

(** Doing [apply HP'] above fails with the following error:

     Error: Impossible to unify "?175" with "y".

    In this case there is an easy fix: doing [destruct HP] _before_
    doing [eapply HQ]. *)

Abort.

Lemma silly2_fixed :
  forall (P : nat -> nat -> Prop) (Q : nat -> Prop),
  (exists y, P 42 y) ->
  (forall x y : nat, P x y -> Q x) ->
  Q 42.
Proof.
  intros P Q HP HQ. destruct HP as [y HP'].
  eapply HQ. apply HP'.
Qed.

(** The [apply HP'] in the last step unifies the existential variable
    in the goal with the variable [y].

    Note that the [assumption] tactic doesn't work in this case, since
    it cannot handle existential variables.  However, Coq also
    provides an [eassumption] tactic that solves the goal if one of
    the premises matches the goal up to instantiations of existential
    variables. We can use it instead of [apply HP'] if we like. *)

Lemma silly2_eassumption : forall (P : nat -> nat -> Prop) (Q : nat -> Prop),
  (exists y, P 42 y) ->
  (forall x y : nat, P x y -> Q x) ->
  Q 42.
Proof.
  intros P Q HP HQ. destruct HP as [y HP']. eapply HQ. eassumption.
Qed.

(** **** Exercise: 2 stars (hoare_asgn_examples_2)  *)
(** Translate these informal Hoare triples...

       {{ X + 1 <= 5 }}  X ::= X + 1  {{ X <= 5 }}
       {{ 0 <= 3 /\ 3 <= 5 }}  X ::= 3  {{ 0 <= X /\ X <= 5 }}

   ...into formal statements (name them [assn_sub_ex1'] and
   [assn_sub_ex2']) and use [hoare_asgn] and [hoare_consequence_pre]
   to prove them. *)
Theorem assn_sub_ex1':
  {{fun st=> st X + 1 <= 5 }}
  (X ::= X + 1)
  {{fun st=> st X <= 5 }}.
Proof.
  eapply  hoare_consequence_pre.
  apply hoare_asgn.
  intros st H.
  unfold assn_sub. simpl. apply H.
Qed.

Theorem assn_sub_ex2':
  {{fun st=> 0 <= 3 /\ 3 <= 5 }}
  (X ::= 3)
  {{fun st=> 0 <= st X /\ st X <= 5 }}.
Proof.
  eapply hoare_consequence_pre.
  apply hoare_asgn.
  intros st H.
  unfold assn_sub. apply H.
Qed.
(* FILL IN HERE *)
(** [] *)

(* ================================================================= *)
(** ** Skip *)

(** Since [SKIP] doesn't change the state, it preserves any
    assertion [P]:

      --------------------  (hoare_skip)
      {{ P }} SKIP {{ P }}
*)

Theorem hoare_skip : forall P,
     {{P}} SKIP {{P}}.
Proof.
  intros P st st' H HP. inversion H. subst.
  assumption.  Qed.

(* ================================================================= *)
(** ** Sequencing *)

(** More interestingly, if the command [c1] takes any state where
    [P] holds to a state where [Q] holds, and if [c2] takes any
    state where [Q] holds to one where [R] holds, then doing [c1]
    followed by [c2] will take any state where [P] holds to one
    where [R] holds:

        {{ P }} c1 {{ Q }}
        {{ Q }} c2 {{ R }}
       ---------------------  (hoare_seq)
       {{ P }} c1;;c2 {{ R }}
*)

Theorem hoare_seq : forall P Q R c1 c2,
     {{Q}} c2 {{R}} ->
     {{P}} c1 {{Q}} ->
     {{P}} c1;;c2 {{R}}.
Proof.
  intros P Q R c1 c2 H1 H2 st st' H12 Pre.
  inversion H12; subst.
  apply (H1 st'0 st'); try assumption.
  apply (H2 st st'0); assumption. Qed.

(** Note that, in the formal rule [hoare_seq], the premises are
    given in backwards order ([c2] before [c1]).  This matches the
    natural flow of information in many of the situations where we'll
    use the rule, since the natural way to construct a Hoare-logic
    proof is to begin at the end of the program (with the final
    postcondition) and push postconditions backwards through commands
    until we reach the beginning. *)

(** Informally, a nice way of displaying a proof using the sequencing
    rule is as a "decorated program" where the intermediate assertion
    [Q] is written between [c1] and [c2]:

      {{ a = n }}
    X ::= a;;
      {{ X = n }}    <---- decoration for Q
    SKIP
      {{ X = n }}
*)

(** Here's an example of a program involving both assignment and
    sequencing. *)

Example hoare_asgn_example3 : forall a n,
  {{fun st => aeval st a = n}}
  (X ::= a;; SKIP)
  {{fun st => st X = n}}.
Proof.
  intros a n. eapply hoare_seq.
  - (* right part of seq *)
    apply hoare_skip.
  - (* left part of seq *)
    eapply hoare_consequence_pre. apply hoare_asgn.
    intros st H. subst. reflexivity.
Qed.

(** We typically use [hoare_seq] in conjunction with
    [hoare_consequence_pre] and the [eapply] tactic, as in this
    example. *)

(** **** Exercise: 2 stars, recommended (hoare_asgn_example4)  *)
(** Translate this "decorated program" into a formal proof:

                   {{ True }} ->>
                   {{ 1 = 1 }}
    X ::= 1;;
                   {{ X = 1 }} ->>
                   {{ X = 1 /\ 2 = 2 }}
    Y ::= 2
                   {{ X = 1 /\ Y = 2 }}

   (Note the use of "[->>]" decorations, each marking a use of
   [hoare_consequence_pre].) *)

Example hoare_asgn_example4 :
  {{fun st => True}} (X ::= 1;; Y ::= 2)
  {{fun st => st X = 1 /\ st Y = 2}}.
Proof.
  eapply hoare_seq.
  apply hoare_asgn.
  eapply hoare_consequence_pre.
  apply hoare_asgn.
  intros st H. unfold assn_sub. simpl. split;reflexivity.

Qed.
(** [] *)

(** **** Exercise: 3 stars (swap_exercise)  *)
(** Write an Imp program [c] that swaps the values of [X] and [Y] and
    show that it satisfies the following specification:

      {{X <= Y}} c {{Y <= X}}

    Your proof should not need to use [unfold hoare_triple]. *)

Definition swap_program : com :=
  Z::=X;;X::=Y;;Y::=Z.

Theorem swap_exercise :
  {{fun st => st X <= st Y}}
  swap_program
  {{fun st => st Y <= st X}}.
Proof.
  eapply hoare_seq.
  eapply hoare_seq.
  apply hoare_asgn.
  apply hoare_asgn.
  eapply hoare_consequence_pre.
  apply hoare_asgn.
  unfold assn_sub. simpl.
  intros st H. apply  H.
Qed.
(** [] *)

(** **** Exercise: 3 stars (hoarestate1)  *)
(** Explain why the following proposition can't be proven:

      forall (a : aexp) (n : nat),
         {{fun st => aeval st a = n}}
           (X ::= 3;; Y ::= a)
         {{fun st => st Y = n}}.
*)

(*
  If a=X;
  after the com, X=3,Y=a=X=3. n is lost.
 *)
(** [] *)

(* ================================================================= *)
(** ** Conditionals *)

(** What sort of rule do we want for reasoning about conditional
    commands?

    Certainly, if the same assertion [Q] holds after executing
    either of the branches, then it holds after the whole conditional.
    So we might be tempted to write:

              {{P}} c1 {{Q}}
              {{P}} c2 {{Q}}
      --------------------------------
      {{P}} IFB b THEN c1 ELSE c2 {{Q}}
*)

(** However, this is rather weak. For example, using this rule,
   we cannot show

     {{ True }}
     IFB X = 0
     THEN Y ::= 2
     ELSE Y ::= X + 1
     FI
     {{ X <= Y }}

   since the rule tells us nothing about the state in which the
   assignments take place in the "then" and "else" branches. *)

(** Fortunately, we can say something more precise.  In the
    "then" branch, we know that the boolean expression [b] evaluates to
    [true], and in the "else" branch, we know it evaluates to [false].
    Making this information available in the premises of the rule gives
    us more information to work with when reasoning about the behavior
    of [c1] and [c2] (i.e., the reasons why they establish the
    postcondition [Q]). *)
(**

              {{P /\  b}} c1 {{Q}}
              {{P /\ ~b}} c2 {{Q}}
      ------------------------------------  (hoare_if)
      {{P}} IFB b THEN c1 ELSE c2 FI {{Q}}
*)

(** To interpret this rule formally, we need to do a little work.
    Strictly speaking, the assertion we've written, [P /\ b], is the
    conjunction of an assertion and a boolean expression -- i.e., it
    doesn't typecheck.  To fix this, we need a way of formally
    "lifting" any bexp [b] to an assertion.  We'll write [bassn b] for
    the assertion "the boolean expression [b] evaluates to [true] (in
    the given state)." *)

Definition bassn b : Assertion :=
  fun st => (beval st b = true).

(** A couple of useful facts about [bassn]: *)

Lemma bexp_eval_true : forall b st,
  beval st b = true -> (bassn b) st.
Proof.
  intros b st Hbe.
  unfold bassn. assumption.  Qed.

Lemma bexp_eval_false : forall b st,
  beval st b = false -> ~ ((bassn b) st).
Proof.
  intros b st Hbe contra.
  unfold bassn in contra.
  rewrite -> contra in Hbe. inversion Hbe.  Qed.

(** Now we can formalize the Hoare proof rule for conditionals
    and prove it correct. *)

Theorem hoare_if : forall P Q b c1 c2,
  {{fun st => P st /\ bassn b st}} c1 {{Q}} ->
  {{fun st => P st /\ ~(bassn b st)}} c2 {{Q}} ->
  {{P}} (IFB b THEN c1 ELSE c2 FI) {{Q}}.
Proof.
  intros P Q b c1 c2 HTrue HFalse st st' HE HP.
  inversion HE; subst.
  - eapply HTrue. apply H5. unfold bassn.
    split. assumption. assumption.
  - eapply HFalse. apply H5. unfold not. unfold bassn. split.
    assumption. intros G. rewrite H4 in G. inversion G. Qed.

(* ----------------------------------------------------------------- *)
(** *** Example *)

(** Here is a formal proof that the program we used to motivate the
    rule satisfies the specification we gave. *)

Example if_example :
    {{fun st => True}}
  IFB X = 0
    THEN Y ::= 2
    ELSE Y ::= X + 1
  FI
    {{fun st => st X <= st Y}}.
Proof.
  (* WORKED IN CLASS *)
  apply hoare_if.
  - (* Then *)
    eapply hoare_consequence_pre. apply hoare_asgn.
    unfold bassn, assn_sub, t_update, assert_implies.
    simpl. intros st [_ H].
    apply beq_nat_true in H.
    rewrite H. omega.
  - (* Else *)
    eapply hoare_consequence_pre. apply hoare_asgn.
    unfold assn_sub, t_update, assert_implies.
    simpl; intros st _. omega.
Qed.

(** **** Exercise: 2 stars (if_minus_plus)  *)
(** Prove the following hoare triple using [hoare_if].  Do not
    use [unfold hoare_triple].  *)

Theorem if_minus_plus :
  {{fun st => True}}
  IFB X <= Y
    THEN Z ::= Y - X
    ELSE Y ::= X + Z
  FI
  {{fun st => st Y = st X + st Z}}.
Proof.
  apply hoare_if.
  - eapply hoare_consequence_pre. apply hoare_asgn.
    unfold bassn, assn_sub, t_update, assert_implies.
    simpl; intros. destruct H. apply leb_complete in H0. omega.
  - eapply hoare_consequence_pre. apply hoare_asgn.
    unfold bassn, assn_sub, t_update, assert_implies.
    simpl; intros. destruct H.
    apply not_true_is_false in H0. apply leb_complete_conv in H0.
    omega.

Qed.
(** [] *)

(* ----------------------------------------------------------------- *)
(** *** Exercise: One-sided conditionals *)

(** **** Exercise: 4 stars (if1_hoare)  *)
(** In this exercise we consider extending Imp with "one-sided
    conditionals" of the form [IF1 b THEN c FI]. Here [b] is a boolean
    expression, and [c] is a command. If [b] evaluates to [true], then
    command [c] is evaluated. If [b] evaluates to [false], then [IF1 b
    THEN c FI] does nothing.

    We recommend that you complete this exercise before attempting the
    ones that follow, as it should help solidify your understanding of
    the material. *)

(** The first step is to extend the syntax of commands and introduce
    the usual notations.  (We've done this for you.  We use a separate
    module to prevent polluting the global name space.) *)

Module If1.

Inductive com : Type :=
  | CSkip : com
  | CAss : string -> aexp -> com
  | CSeq : com -> com -> com
  | CIf : bexp -> com -> com -> com
  | CWhile : bexp -> com -> com
  | CIf1 : bexp -> com -> com.

Notation "'SKIP'" :=
  CSkip.
Notation "c1 ;; c2" :=
  (CSeq c1 c2) (at level 80, right associativity).
Notation "X '::=' a" :=
  (CAss X a) (at level 60).
Notation "'WHILE' b 'DO' c 'END'" :=
  (CWhile b c) (at level 80, right associativity).
Notation "'IFB' e1 'THEN' e2 'ELSE' e3 'FI'" :=
  (CIf e1 e2 e3) (at level 80, right associativity).
Notation "'IF1' b 'THEN' c 'FI'" :=
  (CIf1 b c) (at level 80, right associativity).

(** Next we need to extend the evaluation relation to accommodate
    [IF1] branches.  This is for you to do... What rule(s) need to be
    added to [ceval] to evaluate one-sided conditionals? *)

Reserved Notation "c1 '/' st '\\' st'" (at level 40, st at level 39).

Inductive ceval : com -> state -> state -> Prop :=
  | E_Skip : forall st : state, SKIP / st \\ st
  | E_Ass : forall (st : state) (a1 : aexp) (n : nat) (X : string),
            aeval st a1 = n -> (X ::= a1) / st \\ st & { X --> n }
  | E_Seq : forall (c1 c2 : com) (st st' st'' : state),
            c1 / st \\ st' -> c2 / st' \\ st'' -> (c1 ;; c2) / st \\ st''
  | E_IfTrue : forall (st st' : state) (b1 : bexp) (c1 c2 : com),
               beval st b1 = true ->
               c1 / st \\ st' -> (IFB b1 THEN c1 ELSE c2 FI) / st \\ st'
  | E_IfFalse : forall (st st' : state) (b1 : bexp) (c1 c2 : com),
                beval st b1 = false ->
                c2 / st \\ st' -> (IFB b1 THEN c1 ELSE c2 FI) / st \\ st'
  | E_WhileFalse : forall (b1 : bexp) (st : state) (c1 : com),
                 beval st b1 = false -> (WHILE b1 DO c1 END) / st \\ st
  | E_WhileTrue : forall (st st' st'' : state) (b1 : bexp) (c1 : com),
                  beval st b1 = true ->
                  c1 / st \\ st' ->
                  (WHILE b1 DO c1 END) / st' \\ st'' ->
                  (WHILE b1 DO c1 END) / st \\ st''
  | E_If1True: forall (st st':state) (b:bexp) (c1:com),
                  beval st b = true->
                  c1 / st \\ st'-> (IF1 b THEN c1 FI) /st \\ st'
  | E_If1False: forall (st :state) (b:bexp) (c1:com),
                  beval st b = false->
                  (IF1 b THEN c1 FI) /st \\ st
  where "c1 '/' st '\\' st'" := (ceval c1 st st').

(** Now we repeat (verbatim) the definition and notation of Hoare triples. *)

Definition hoare_triple (P:Assertion) (c:com) (Q:Assertion) : Prop :=
  forall st st',
       c / st \\ st'  ->
       P st  ->
       Q st'.

Notation "{{ P }}  c  {{ Q }}" := (hoare_triple P c Q)
                                  (at level 90, c at next level)
                                  : hoare_spec_scope.

(** Finally, we (i.e., you) need to state and prove a theorem,
    [hoare_if1], that expresses an appropriate Hoare logic proof rule
    for one-sided conditionals. Try to come up with a rule that is
    both sound and as precise as possible. *)

Theorem hoare_if1 : forall P Q b c1,
  {{fun st => P st /\ bassn b st}} c1 {{Q}} ->
  {{fun st => P st /\ ~(bassn b st)}} SKIP {{Q}} ->
  {{P}} (IF1 b THEN c1 FI) {{Q}}.
Proof.
  intros. unfold hoare_triple;intros.
  inversion H1; subst.
  - eapply H. apply H8. split. apply H2. apply bexp_eval_true. apply H5.
  - eapply H0. apply E_Skip. split. apply H2.
    unfold not. intros Z. apply bexp_eval_false in H7. unfold not in H7.
    apply H7 in Z. inversion Z.

Qed.

(** For full credit, prove formally [hoare_if1_good] that your rule is
    precise enough to show the following valid Hoare triple:

  {{ X + Y = Z }}
  IF1 !(Y = 0) THEN
    X ::= X + Y
  FI
  {{ X = Z }}
*)

Example hoare_if1_good:
  {{fun st=> (st X) + (st Y) = (st Z) }}
  IF1 !(Y = 0) THEN
    X ::= X + Y
  FI
  {{fun st=> st X = st Z }}.
Proof.
  apply hoare_if1; unfold hoare_triple;intros.
  - destruct H0. inversion H;subst.
    rewrite t_update_eq. simpl. rewrite t_update_neq.
    apply H0. destruct (beq_string X Z) eqn:T.
    + inversion T.
    + apply beq_string_false_iff in T. apply T.
  - destruct H0. inversion H;subst.
    unfold bassn in H1. apply not_true_is_false in H1. simpl in H1.
    apply negb_false_iff in H1. apply beq_nat_true in H1.
    rewrite H1 in H0. rewrite (plus_comm (st' X) 0) in H0.
    rewrite plus_0_l in H0. apply H0.
Qed.

(** Hint: Your proof of this triple may need to use the other proof
    rules also. Because we're working in a separate module, you'll
    need to copy here the rules you find necessary. *)

End If1.
(** [] *)

(* ================================================================= *)
(** ** Loops *)

(** Finally, we need a rule for reasoning about while loops. *)

(** Suppose we have a loop

      WHILE b DO c END

    and we want to find a pre-condition [P] and a post-condition
    [Q] such that

      {{P}} WHILE b DO c END {{Q}}

    is a valid triple. *)

(** First of all, let's think about the case where [b] is false at the
    beginning -- i.e., let's assume that the loop body never executes
    at all.  In this case, the loop behaves like [SKIP], so we might
    be tempted to write: *)

(**

      {{P}} WHILE b DO c END {{P}}.
*)

(** But, as we remarked above for the conditional, we know a
    little more at the end -- not just [P], but also the fact
    that [b] is false in the current state.  So we can enrich the
    postcondition a little: *)
(**

      {{P}} WHILE b DO c END {{P /\ ~b}}
*)

(** What about the case where the loop body _does_ get executed?
    In order to ensure that [P] holds when the loop finally
    exits, we certainly need to make sure that the command [c]
    guarantees that [P] holds whenever [c] is finished.
    Moreover, since [P] holds at the beginning of the first
    execution of [c], and since each execution of [c]
    re-establishes [P] when it finishes, we can always assume
    that [P] holds at the beginning of [c].  This leads us to the
    following rule: *)
(**

                   {{P}} c {{P}}
        -----------------------------------
        {{P}} WHILE b DO c END {{P /\ ~b}}
*)
(** This is almost the rule we want, but again it can be improved a
    little: at the beginning of the loop body, we know not only that
    [P] holds, but also that the guard [b] is true in the current
    state. *)

(** This gives us a little more information to use in
    reasoning about [c] (showing that it establishes the invariant by
    the time it finishes).  *)
(** This gives us the final version of the rule: *)
(**

               {{P /\ b}} c {{P}}
        -----------------------------------  (hoare_while)
        {{P}} WHILE b DO c END {{P /\ ~b}}

    The proposition [P] is called an _invariant_ of the loop.
*)

Theorem hoare_while : forall P b c,
  {{fun st => P st /\ bassn b st}} c {{P}} ->
  {{P}} WHILE b DO c END {{fun st => P st /\ ~ (bassn b st)}}.
Proof.
  intros P b c Hhoare st st' He HP.
  (* Like we've seen before, we need to reason by induction
     on [He], because, in the "keep looping" case, its hypotheses
     talk about the whole loop instead of just [c]. *)
  remember (WHILE b DO c END) as wcom eqn:Heqwcom.
  induction He;
    try (inversion Heqwcom); subst; clear Heqwcom.
  - (* E_WhileFalse *)
    split. assumption. apply bexp_eval_false. assumption.
  - (* E_WhileTrue *)
    apply IHHe2. reflexivity.
    apply (Hhoare st st'). assumption.
      split. assumption. apply bexp_eval_true. assumption.
Qed.

(** One subtlety in the terminology is that calling some assertion [P]
    a "loop invariant" doesn't just mean that it is preserved by the
    body of the loop in question (i.e., [{{P}} c {{P}}], where [c] is
    the loop body), but rather that [P] _together with the fact that
    the loop's guard is true_ is a sufficient precondition for [c] to
    ensure [P] as a postcondition.

    This is a slightly (but importantly) weaker requirement.  For
    example, if [P] is the assertion [X = 0], then [P] _is_ an
    invariant of the loop

    WHILE X = 2 DO X := 1 END

    although it is clearly _not_ preserved by the body of the
    loop. *)

Example while_example :
    {{fun st => st X <= 3}}
  WHILE X <= 2
  DO X ::= X + 1 END
    {{fun st => st X = 3}}.
Proof.
  eapply hoare_consequence_post.
  apply hoare_while.
  eapply hoare_consequence_pre.
  apply hoare_asgn.
  unfold bassn, assn_sub, assert_implies, t_update. simpl.
    intros st [H1 H2]. apply leb_complete in H2. omega.
  unfold bassn, assert_implies. intros st [Hle Hb].
    simpl in Hb. destruct (leb (st X) 2) eqn : Heqle.
    exfalso. apply Hb; reflexivity.
    apply leb_iff_conv in Heqle. omega.
Qed.
(** We can use the WHILE rule to prove the following Hoare triple... *)

Theorem always_loop_hoare : forall P Q,
  {{P}} WHILE true DO SKIP END {{Q}}.
Proof.
  (* WORKED IN CLASS *)
  intros P Q.
  apply hoare_consequence_pre with (P' := fun st : state => True).
  eapply hoare_consequence_post.
  apply hoare_while.
  - (* Loop body preserves invariant *)
    apply hoare_post_true. intros st. apply I.
  - (* Loop invariant and negated guard imply postcondition *)
    simpl. intros st [Hinv Hguard].
    exfalso. apply Hguard. reflexivity.
  - (* Precondition implies invariant *)
    intros st H. constructor.  Qed.

(** Of course, this result is not surprising if we remember that
    the definition of [hoare_triple] asserts that the postcondition
    must hold _only_ when the command terminates.  If the command
    doesn't terminate, we can prove anything we like about the
    post-condition. *)

(** Hoare rules that only talk about what happens when commands
    terminate (without proving that they do) are often said to
    describe a logic of "partial" correctness.  It is also possible to
    give Hoare rules for "total" correctness, which build in the fact
    that the commands terminate. However, in this course we will only
    talk about partial correctness. *)

(* ----------------------------------------------------------------- *)
(** *** Exercise: [REPEAT] *)

(** **** Exercise: 4 stars, advanced (hoare_repeat)  *)
(** In this exercise, we'll add a new command to our language of
    commands: [REPEAT] c [UNTIL] a [END]. You will write the
    evaluation rule for [repeat] and add a new Hoare rule to the
    language for programs involving it.  (You may recall that the
    evaluation rule is given in an example in the [Auto] chapter.
    Try to figure it out yourself here rather than peeking.) *)

Module RepeatExercise.

Inductive com : Type :=
  | CSkip : com
  | CAsgn : string -> aexp -> com
  | CSeq : com -> com -> com
  | CIf : bexp -> com -> com -> com
  | CWhile : bexp -> com -> com
  | CRepeat : com -> bexp -> com.

(** [REPEAT] behaves like [WHILE], except that the loop guard is
    checked _after_ each execution of the body, with the loop
    repeating as long as the guard stays _false_.  Because of this,
    the body will always execute at least once. *)

Notation "'SKIP'" :=
  CSkip.
Notation "c1 ;; c2" :=
  (CSeq c1 c2) (at level 80, right associativity).
Notation "X '::=' a" :=
  (CAsgn X a) (at level 60).
Notation "'WHILE' b 'DO' c 'END'" :=
  (CWhile b c) (at level 80, right associativity).
Notation "'IFB' e1 'THEN' e2 'ELSE' e3 'FI'" :=
  (CIf e1 e2 e3) (at level 80, right associativity).
Notation "'REPEAT' e1 'UNTIL' b2 'END'" :=
  (CRepeat e1 b2) (at level 80, right associativity).

(** Add new rules for [REPEAT] to [ceval] below.  You can use the rules
    for [WHILE] as a guide, but remember that the body of a [REPEAT]
    should always execute at least once, and that the loop ends when
    the guard becomes true. *)

Inductive ceval : state -> com -> state -> Prop :=
  | E_Skip : forall st,
      ceval st SKIP st
  | E_Ass  : forall st a1 n X,
      aeval st a1 = n ->
      ceval st (X ::= a1) (st & { X --> n })
  | E_Seq : forall c1 c2 st st' st'',
      ceval st c1 st' ->
      ceval st' c2 st'' ->
      ceval st (c1 ;; c2) st''
  | E_IfTrue : forall st st' b1 c1 c2,
      beval st b1 = true ->
      ceval st c1 st' ->
      ceval st (IFB b1 THEN c1 ELSE c2 FI) st'
  | E_IfFalse : forall st st' b1 c1 c2,
      beval st b1 = false ->
      ceval st c2 st' ->
      ceval st (IFB b1 THEN c1 ELSE c2 FI) st'
  | E_WhileFalse : forall b1 st c1,
      beval st b1 = false ->
      ceval st (WHILE b1 DO c1 END) st
  | E_WhileTrue : forall st st' st'' b1 c1,
      beval st b1 = true ->
      ceval st c1 st' ->
      ceval st' (WHILE b1 DO c1 END) st'' ->
      ceval st (WHILE b1 DO c1 END) st''
  | E_NoRepeat: forall st st' b1 e1,
      ceval st e1 st'->
      beval st' b1 = true ->
      ceval st (REPEAT e1 UNTIL b1 END) st'
  | E_Repeat: forall e1 st st' st'' b1,
      ceval st e1 st' ->
      beval st' b1 = false ->
      ceval st' (REPEAT e1 UNTIL b1 END) st''->
      ceval st  (REPEAT e1 UNTIL b1 END) st''
(* FILL IN HERE *)
.

(** A couple of definitions from above, copied here so they use the
    new [ceval]. *)

Notation "c1 '/' st '\\' st'" := (ceval st c1 st')
                                 (at level 40, st at level 39).

Definition hoare_triple (P:Assertion) (c:com) (Q:Assertion)
                        : Prop :=
  forall st st', (c / st \\ st') -> P st -> Q st'.

Notation "{{ P }}  c  {{ Q }}" :=
  (hoare_triple P c Q) (at level 90, c at next level).

(** To make sure you've got the evaluation rules for [REPEAT] right,
    prove that [ex1_repeat] evaluates correctly. *)

Definition ex1_repeat :=
  REPEAT
    X ::= 1;;
    Y ::= Y + 1
  UNTIL X = 1 END.

Theorem ex1_repeat_works :
  ex1_repeat / { --> 0 } \\ { X --> 1 ; Y --> 1 }.
Proof.
  apply E_NoRepeat.
  - eapply E_Seq. apply E_Ass.  reflexivity.
    simpl. eapply E_Ass. reflexivity.
  - simpl. reflexivity.

Qed.

(** Now state and prove a theorem, [hoare_repeat], that expresses an
    appropriate proof rule for [repeat] commands.  Use [hoare_while]
    as a model, and try to make your rule as precise as possible. *)

Theorem hoare_repeat: forall P Q b c,
  {{fun st => Q st /\ ~ bassn b st}} c {{Q}} ->
  {{P}} c {{Q}} ->
  {{P}} REPEAT c UNTIL b END {{fun st => Q st /\ bassn b st}}.
Proof.

  unfold hoare_triple; intros. generalize dependent P.
  remember (REPEAT c UNTIL b END) as rcom eqn:Req.
  induction H1;inversion Req;subst.
  - intros. split. eapply H2. apply H1. apply H3. apply bexp_eval_true. apply H0.
  - intros. eapply IHceval2. reflexivity.
    apply H. simpl. split.
    eapply H1. apply H1_. apply H2.  apply bexp_eval_false. apply H0.
Qed.

(** For full credit, make sure (informally) that your rule can be used
    to prove the following valid Hoare triple:

  {{ X > 0 }}
  REPEAT
    Y ::= X;;
    X ::= X - 1
  UNTIL X = 0 END
  {{ X = 0 /\ Y > 0 }}
*)(*
Example hoare_repeat_exam1 :
  {{ fun st => st X > 0 }}
  REPEAT
    Y ::= AId X;;
    X ::= AMinus (AId X) (ANum 1)
  UNTIL BEq (AId X) (ANum 0) END
  {{ fun st => st X = 0 /\ st Y > 0 }}.
Proof.
  destruct (beq_string X Y) eqn:eqxy; inversion eqxy.
  intros st st' H1 H2.
  assert (((fun st' => st' Y > 0 /\ st' Y = st' X + 1) st'  /\ bassn (BEq (AId X) (ANum 0)) st') -> st' X = 0 /\ st' Y > 0).
    intros [[Ha1 Ha2] Ha3].
    inversion Ha3. destruct (st' X). apply conj; auto.
      inversion H0.
  apply H.
  assert({{ fun st => st X > 0 }}
  REPEAT
    Y ::= AId X;;
    X ::= AMinus (AId X) (ANum 1)
  UNTIL BEq (AId X) (ANum 0) END
  {{fun st => (fun st' : string -> nat => st' Y > 0 /\ st' Y = st' X + 1) st /\
    bassn (BEq (AId X) (ANum 0)) st}}).
  apply hoare_repeat.
-  unfold hoare_triple; intros.
  inversion H0; subst. inversion H7; subst.
  inversion H9; subst.
  rewrite update_neq; auto. rewrite update_eq.
  destruct (st0 X) eqn:stx. inversion H3 as [H31 H32].
  assert (False). apply H32. apply bexp_eval_true. simpl. rewrite stx. auto.
  inversion H4. simpl; apply conj. unfold gt. rewrite stx. apply lt_0_Sn.
  rewrite update_eq. rewrite update_neq; auto.
  rewrite stx. simpl. rewrite <- minus_n_O. rewrite plus_comm.
  reflexivity. unfold hoare_triple. intros.
  inversion H0; subst. inversion H7; subst. inversion H9; subst.
  rewrite update_neq; auto. rewrite update_eq; auto. apply conj.
  simpl. assumption. simpl. rewrite update_eq. rewrite update_neq; auto.
  destruct (st0 X). inversion H3. simpl. rewrite <- minus_n_O. rewrite plus_comm.
  reflexivity. unfold hoare_triple in H0.
  eapply H0. apply H1. assumption.
  Qed.

1 subgoal
eqxy : beq_string X Y = false
st, st' : state
H1 : (REPEAT Y ::= X;; X ::= X - 1
      UNTIL X = 0 END) / st \\ st'
H2 : st X > 0
H : (fun st' : string -> nat =>
     st' Y > 0 /\ st' Y = st' X + 1) st' /\
    bassn (X = 0) st' ->
    st' X = 0 /\ st' Y > 0
______________________________________(1/1)
{{fun st0 : state =>
  (st0 Y > 0 /\ st0 Y = st0 X + 1) /\
  ~ bassn (X = 0) st0}}
Y ::= X;; X ::= X - 1
{{fun st0 : state =>
  st0 Y > 0 /\ st0 Y = st0 X + 1}}

1 subgoal
eqxy : beq_string X Y = false
st, st' : state
H1 : (REPEAT Y ::= X;; X ::= X - 1
      UNTIL X = 0 END) / st \\ st'
H2 : st X > 0
H : (fun st' : string -> nat =>
     st' Y > 0 /\ st' Y = st' X + 1) st' /\
    bassn (X = 0) st' ->
    st' X = 0 /\ st' Y > 0
______________________________________(1/1)
(st' Y > 0 /\ st' Y = st' X + 1) /\
bassn (X = 0) st'

*)
Example hoare_repeat_example:
  {{fun st=>st X > 0 }}
  REPEAT
    Y ::= X;;
    X ::= X - 1
  UNTIL X = 0 END
  {{fun st=>st X = 0 /\st Y > 0 }}.
Proof.
  destruct (beq_string X Y) eqn:eqxy;
  try(apply beq_string_false_iff in eqxy);
  try(apply beq_string_true_iff in eqxy). inversion eqxy.
  intros st st' H1 H2.
  assert (((fun st' => st' Y > 0 /\ st' Y = st' X + 1) st'  /\ bassn (X=0) st' ) -> st' X = 0 /\ st' Y > 0).
  { intros;destruct H;destruct H;inversion H0;apply beq_nat_true in H5.
    split;omega. }

  apply H.
  assert({{ fun st => st X > 0 }}
  REPEAT
    Y ::= X;;
    X ::= X-1
  UNTIL X=0 END
  {{fun st => (fun st'=>st' Y > 0 /\ st' Y = st' X + 1) st /\ bassn (X=0) st}}).
  { apply hoare_repeat.
    - unfold hoare_triple; intros.
      inversion H0; subst. inversion H7; subst. inversion H9; subst.
      rewrite t_update_neq. rewrite t_update_eq. simpl.
      destruct (st0 X) eqn:stx.
      + inversion H3. exfalso. unfold not,bassn in H5.
        simpl in H5. rewrite stx in H5. apply H5. reflexivity.
      + simpl. split. omega.
        rewrite t_update_eq. rewrite t_update_neq. omega. auto.
      + assumption.
    -  unfold hoare_triple. intros.
       inversion H0; subst. inversion H7; subst. inversion H9; subst.
       rewrite t_update_neq; auto. rewrite t_update_eq; auto. split.
       + simpl. assumption.
       + simpl. rewrite t_update_eq. rewrite t_update_neq; auto. omega. }
   unfold hoare_triple in H0.
    eapply H0. apply H1. assumption.
Qed.

End RepeatExercise.
(** [] *)

(* ################################################################# *)
(** * Summary *)

(** So far, we've introduced Hoare Logic as a tool for reasoning about
    Imp programs. The rules of Hoare Logic are:

             ------------------------------ (hoare_asgn)
             {{Q [X |-> a]}} X::=a {{Q}}

             --------------------  (hoare_skip)
             {{ P }} SKIP {{ P }}

               {{ P }} c1 {{ Q }}
               {{ Q }} c2 {{ R }}
              ---------------------  (hoare_seq)
              {{ P }} c1;;c2 {{ R }}

              {{P /\  b}} c1 {{Q}}
              {{P /\ ~b}} c2 {{Q}}
      ------------------------------------  (hoare_if)
      {{P}} IFB b THEN c1 ELSE c2 FI {{Q}}

               {{P /\ b}} c {{P}}
        -----------------------------------  (hoare_while)
        {{P}} WHILE b DO c END {{P /\ ~b}}

                {{P'}} c {{Q'}}
                   P ->> P'
                   Q' ->> Q
         -----------------------------   (hoare_consequence)
                {{P}} c {{Q}}

    In the next chapter, we'll see how these rules are used to prove
    that programs satisfy specifications of their behavior. *)

(* ################################################################# *)
(** * Additional Exercises *)

(** **** Exercise: 3 stars (hoare_havoc)  *)
(** In this exercise, we will derive proof rules for a [HAVOC]
    command, which is similar to the nondeterministic [any] expression
    from the the [Imp] chapter.

    First, we enclose this work in a separate module, and recall the
    syntax and big-step semantics of Himp commands. *)

Module Himp.

Inductive com : Type :=
  | CSkip : com
  | CAsgn : string -> aexp -> com
  | CSeq : com -> com -> com
  | CIf : bexp -> com -> com -> com
  | CWhile : bexp -> com -> com
  | CHavoc : string -> com.

Notation "'SKIP'" :=
  CSkip.
Notation "X '::=' a" :=
  (CAsgn X a) (at level 60).
Notation "c1 ;; c2" :=
  (CSeq c1 c2) (at level 80, right associativity).
Notation "'WHILE' b 'DO' c 'END'" :=
  (CWhile b c) (at level 80, right associativity).
Notation "'IFB' e1 'THEN' e2 'ELSE' e3 'FI'" :=
  (CIf e1 e2 e3) (at level 80, right associativity).
Notation "'HAVOC' X" := (CHavoc X) (at level 60).

Reserved Notation "c1 '/' st '\\' st'" (at level 40, st at level 39).

Inductive ceval : com -> state -> state -> Prop :=
  | E_Skip : forall st : state, SKIP / st \\ st
  | E_Ass : forall (st : state) (a1 : aexp) (n : nat) (X : string),
            aeval st a1 = n -> (X ::= a1) / st \\ st & { X --> n }
  | E_Seq : forall (c1 c2 : com) (st st' st'' : state),
            c1 / st \\ st' -> c2 / st' \\ st'' -> (c1 ;; c2) / st \\ st''
  | E_IfTrue : forall (st st' : state) (b1 : bexp) (c1 c2 : com),
               beval st b1 = true ->
               c1 / st \\ st' -> (IFB b1 THEN c1 ELSE c2 FI) / st \\ st'
  | E_IfFalse : forall (st st' : state) (b1 : bexp) (c1 c2 : com),
                beval st b1 = false ->
                c2 / st \\ st' -> (IFB b1 THEN c1 ELSE c2 FI) / st \\ st'
  | E_WhileFalse : forall (b1 : bexp) (st : state) (c1 : com),
                 beval st b1 = false -> (WHILE b1 DO c1 END) / st \\ st
  | E_WhileTrue : forall (st st' st'' : state) (b1 : bexp) (c1 : com),
                  beval st b1 = true ->
                  c1 / st \\ st' ->
                  (WHILE b1 DO c1 END) / st' \\ st'' ->
                  (WHILE b1 DO c1 END) / st \\ st''
  | E_Havoc : forall (st : state) (X : string) (n : nat),
              (HAVOC X) / st \\ st & { X --> n }

  where "c1 '/' st '\\' st'" := (ceval c1 st st').

(** The definition of Hoare triples is exactly as before. *)

Definition hoare_triple (P:Assertion) (c:com) (Q:Assertion) : Prop :=
  forall st st', c / st \\ st' -> P st -> Q st'.

Notation "{{ P }}  c  {{ Q }}" := (hoare_triple P c Q)
                                  (at level 90, c at next level)
                                  : hoare_spec_scope.

(** Complete the Hoare rule for [HAVOC] commands below by defining
    [havoc_pre] and prove that the resulting rule is correct. *)

Definition havoc_pre (X : string) (Q : Assertion) : Assertion :=
  fun st => forall n, Q (st & {X-->n}).

Theorem hoare_havoc : forall (Q : Assertion) (X : string),
  {{ havoc_pre X Q }} HAVOC X {{ Q }}.
Proof.
  intros.
  unfold hoare_triple. intros.
  unfold havoc_pre in H0. inversion H;subst.
  apply H0.
Qed.

End Himp.
(** [] *)

(** $Date: 2017-12-05 10:28:10 -0600 (Tue, 05 Dec 2017) $ *)


From Coq.Unicode Require Import Utf8.
From Coq.micromega Require Import Lia.
From trillium.prelude Require Import classical sigma.

Definition injective {A B} (f : A → B) := ∀ x y, f x = f y → x = y.

Lemma injective_compose {A B C} (f : A → B) (g : B → C) :
  injective f → injective g → injective (λ x, g (f x)).
Proof. intros Hf Hg x y Hxy; apply Hf, Hg; trivial. Qed.

Definition smaller_card A B := ∀ f : B → A, ¬ injective f.

Section quantifiers.
  Context (A : Type) (R : A → A → Prop)
          (Htotal : ∀ x y, R x y ∨ R y x)
          (Htrans : ∀ x y z, R x y → R y z → R x z).

  Definition subset := A → Prop.

  Definition subset_type (S : subset) : Type := sig S.

  Definition cofinal (S : subset) := ∀ x, ∃ y, S y ∧ R x y.

  Definition regular :=
    ∀ (S : subset), cofinal S → ∃ f : A → (subset_type S), injective f.

  Lemma not_cofinal_smaller S :
    ¬ cofinal S → ∃ x, ∀ y, S y → ¬ R x y.
  Proof.
    intros Hncf.
    apply not_forall_exists_not in Hncf as [x Hncf]; eauto 10.
  Qed.

  Definition union {I : Type} (fam : I → subset) : subset := λ x, ∃ i, fam i x.

  Lemma regular_union_of_smaller (I : Type) (fam : I → subset) :
    regular →
    smaller_card I A →
    (∀ i, ¬ cofinal (fam i)) →
    ¬ (cofinal (union fam)).
  Proof.
    intros Hreg HI Hfam Hcnf.
    pose proof (Hreg _ Hcnf) as [f Hf].
    assert (∀ i, ∃ x, ∀ y, fam i y → ¬ R x y) as Hmax.
    { intros i. apply not_cofinal_smaller; auto. }
    apply Choice in Hmax as [g Hg].
    set (Img := λ x, ∃ i, g i = x).
    assert (∀ x : subset_type Img, ∃ i : I, proj1_sig x = g i) as Himgback.
    { intros [? [? ?]]; eauto. }
    apply Choice in Himgback as [h Hh].
    assert (injective h).
    { intros x y Hxy.
      apply sig_eq; rewrite !Hh, Hxy; trivial. }
    assert (¬ cofinal Img) as Himgncnf.
    { intros Himgcnf.
      destruct (Hreg _ Himgcnf) as [h' Hh'].
      apply (HI (λ x, h (h' x))).
      apply injective_compose; auto. }
    apply not_cofinal_smaller in Himgncnf as [z Hz].
    destruct (Hcnf z) as [u [[i Hu1] Hu2]].
    assert (R z (g i)) as Hzgi.
    { pose proof (Hg i _ Hu1).
      destruct (Htotal (g i) u); [tauto|].
      eapply Htrans; eauto. }
    apply Hz in Hzgi; [|unfold Img]; eauto.
  Qed.

  Lemma regular_union_biger (I : Type) (fam : I → subset) :
    regular →
    smaller_card I A →
    (cofinal (union fam)) →
    ∃ i, cofinal (fam i).
  Proof.
    intros Hreg HI Hcnf.
    cut (∃ i, ¬ ¬ cofinal (fam i)).
    { intros [? ?]; eexists; apply NNP_P; eauto. }
    apply not_forall_exists_not.
    intros Hfam.
    eapply regular_union_of_smaller; eauto.
  Qed.

  Context (B : Type) (P : A → B → Prop).

  Definition fiber (b : B) : subset := λ x, P x b.

  Context (Hmono : ∀ x y z, R x y → P y z → P x z).

  Lemma cofinal_fiber_forall (b : B) : cofinal (fiber b) → ∀ x, P x b.
  Proof.
    intros Hb x.
    destruct (Hb x) as (y & Hy1 & Hy2).
    eapply Hmono; eauto.
  Qed.

  Context (Hrefl : ∀ x, R x x).

  Lemma forall_exists_swap :
    regular →
    smaller_card B A →
    (∀ x : A, ∃ y : B, P x y) → ∃ y, ∀ x, P x y.
  Proof.
    intros Hreg HB Hfe.
    destruct (regular_union_biger _ fiber) as [y Hy]; [trivial|trivial| |].
    { intros x. destruct (Hfe x) as [y Hy].
      exists x; split; [|apply Hrefl].
      eexists; eauto. }
    exists y; apply cofinal_fiber_forall; trivial.
  Qed.

End quantifiers.

Example nat_regular : regular nat le.
Proof.
  intros St HS.
  pose proof (Choice _ _ _ HS) as [f Hf].
  set (g :=
         fix g n :=
         match n with
         | 0 => f 0
         | S n' => f (S (g n'))
         end).
  assert (∀ n, St (g n)) as Hg.
  { destruct n; apply Hf. }
  assert (∀ x y, x < y → g x < g y) as Hglt.
  { intros x y; revert x.
    induction y as [|y IHy]; simpl; [lia|].
    intros x Hx.
    destruct (PeanoNat.Nat.eq_dec x y) as [->|].
    - pose proof (Hf (S (g y))) as [_ ?].
      lia.
    - pose proof (Hf (S (g y))) as [_ ?].
      assert (g x < g y).
      { apply IHy; lia. }
      lia. }
  set (G n := exist _ _ (Hg n)).
  exists G.
  intros x y Hxy.
  assert (proj1_sig (G x) = proj1_sig (G y)) by (rewrite Hxy; auto); simpl in *.
  destruct (Compare_dec.lt_eq_lt_dec x y) as [[|]|]; [|trivial; fail|].
  - assert (g x < g y) by (apply Hglt; lia); lia.
  - assert (g y < g x) by (apply Hglt; lia); lia.
Qed.


(* Definición de la función fst para naturales. *)
Definition fst (n1 n2: nat) : nat := n1.

(* Definición de una función que toma otra función y dos parámetros y devuelve
   la aplicación de la segunda función a los parámetros con órden invertido. *)
Definition swap (c: nat -> nat -> nat) (n1 n2: nat) : nat := c n2 n1.